Title

A Computer Program for the Analysis of Prismatic Solids

Authors

Wilson, Edward

Pretorius, Pieter

Publication Date

1970-09-01

REPORT NO.
UC SESM 70-21

STRUCTURES AND MATERIALS RESEARCH
DEPARTMENT OF CIVIL ENGINEERING

A COMPUTER PROGRAM FOR THE ANALYSIS OF PRISMATIC SOLIDS

by

EDWARD L. WILSON

and

PIETER C. PRETORIUS

SEPTEMBER 1970

STRUCTURAL ENGINEERING LABORATORY
UNIVERSITY OF CALIFORNIA
BERKELEY CALIFORNIA

A COMPUTER PROGRAM FOR THE ANALYSIS OF PRISMATIC SOLIDS

by

Edward L. Wilson

Pieter C. Pretorius

Department of Civil Engineering
University of California
Berkeley, California

September 1970

ABSTRACT

A three-dimensional solid which has a constant two-dimensional geometric shape with respect to an infinite third dimension is defined as a prismatic solid. The typical two-dimensional cross section may have arbitrary geometric shape, material properties, and boundary conditions. The finite element technique, which utilizes an infinitely long element of triangular cross section, it used as the basic method of analysis. The three components of displacement within each element are expanded in terms of trigonometric functions which permits the three-dimensional analysis to be reduced to a series of two-dimensional analyses. The accuracy of the method is illustrated by a comparison with the exact solution of a point load on the half-space. A solution of the very important problem of the point load on the quarter space is also given.

TABLE OF CONTENTS

INTRODUCTION

The determination of stresses and displacements within massive underground structures is of considerable practical interest. Only structures which can be idealized as plate strain or as the half space have been attacked analytically. Important problems such as arbitrary loadings within long tunnels or arbitrary surface loads near the edge of a cliff have not been solved. In the present paper, this general class of three-dimensional problems is attacked by the finite element method, and a digital computer program is developed for structures of arbitrary geometry and loading.

Since the introduction of the finite element method for the solution of plane stress structures [1], the technique has been successfully applied to plates [2], axisymmetric solids [3,4] and shells [5,6], three-dimensional solids [7], torsional behavior of shafts [8], and other nonstructural boundary value problems [9,10]. In the finite element idealization of solids, the continuous body is replaced by a system of elements (subregions) in which the displacement fields are approximate and expressed in terms of the values of the displacements at the node points. Since the node points are common to adjacent elements, a continuous displacement field may be defined over the complete structure. From energy considerations, a linear set of equilibrium equations in terms of the displacements at the nodes (generalized coordinates) are developed for the system. These equations always produce a symmetric, positive-definite matrix and for most problems may be placed in band form. Therefore, a solution may be found with a minimum of computer storage and computational time. In general, the finite element method has many practical advantages when compared with other methods. Structures of nonhomogeneous, anisotropic materials and with arbitrary geometric shapes and boundary conditions are easily represented.

The advantages of the prismatic space formulation as compared to the arbitrary three-dimensional finite element are in respect to accuracy and computational effort. Existing three-dimensional finite element programs are not practical for the prismatic space class of structures because of the tremendous number of unknown displacements required to represent the system. For example, a direct application of a three-dimensional finite element program may require several hours of computer time; whereas, the use of the prismatic space formulation given here may reduce the computational time to several minutes.

METHOD OF ANALYSIS

In the present investigation, the finite element method is applied to the analysis of three-dimensional solids with one infinite dimension and a typical two-dimensional cross-section. This "prismatic space" is illustrated in Figure 1. In this paper, loading which is symmetric with respect to the plane define by $z = 0$ is considered. Other types of loading may be treated by translation of origin and superposition. Approximate solutions to solids for which the loading is not periodic may also be obtained. For periodic loading the length is arbitrary and may be set to a very large value compared to the area of interest near the load. Therefore, the displacements at a distance ℓ from the load may be made to approximate the boundary conditions at infinity.

A. Equilibrium of Complete System

The development of the nodal point equilibrium equations for a prismatic solid is similar in many respects to the procedure used for axisymmetric solids subjected to non-axisymmetric loading[4]. A convenient starting point for the development is the potential energy of the system, which in the notation of Reference [11] is

$$\Phi = \int_{Vol} \frac{1}{2} \varepsilon_i \tau_i \, dV - \int_{Vol} u_i F_i \, dV - \int_{Area} u_i T_i \, dA \qquad (1)$$

Or, if written in matrix form

$$\Phi = \int_{Vol} \frac{1}{2} \underline{\varepsilon}^T \underline{\tau} \, dV - \int_{Vol} \underline{u}^T \underline{F} \, dV - \int_{Area} \underline{u}^T \underline{T} \, dA \qquad (2)$$



FIGURE I THE PRISMATIC SPACE

For a finite element system, which is composed of M arbitrary elements, Equation (2) may be written as a sum of integrals over individual elements.

$$\Phi = \sum_{m=1}^{M} \int_{Vol} \underset{-}{\varepsilon}_m^T \tau_m \, dV_m - \int_{Vol} \underset{-}{u}_m^T \underset{-}{F}_m \, dV_m - \int_{Area} \underset{-}{u}_m^T \underset{-}{T}_m \, dA_m \qquad (3)$$

It is apparent that the surface integral exists only if the m^{th} element is on the boundary and is subjected to surface tractions Tm.

In order to convert the expression for potential energy to one in terms of a finite number of generalized coordinates, it is necessary to assume a solution for the displacement field within each element. If the solution is to converge to the exact solution as the number of elements is increased this assumed displacement field should satisfy compatibility between elements of the system. If the generalized coordinates are selected as the displacement at the node points of the finite element system, then an expression for the displacements $\underset{-}{u}_m$ within each element in terms of generalized coordinates $\underline{U}$ may be developed and written in matrix form as

$$\underset{-}{u}_m = \underset{-}{d}_m \underline{U} \qquad (4)$$

or in transposed form

$$u_m^T = U^T d_m^T \qquad (5)$$

By the use of strain-displacement relations the strains may also by expressed in terms of the generalized coordinates

$$\underset{-}{\varepsilon}_m = \underset{-}{a}_m \underline{U} \qquad (6)$$

Or in transposed form

$$\underset{-}{\varepsilon}_m^T = \underline{U}^T \underset{-}{a}_m^T \tag{7}$$

Within each element an elastic stress-strain relationship must be satisfied and may be written in matrix form as

$$\underset{-}{\tau}_m = \underset{-}{c}_m \underset{-}{\varepsilon}_m \tag{8}$$

If Eqs. (4) through (8) are substituted into Eq. (3) the potential energy for the finite element system is

$$\Phi = \frac{1}{2} \underline{U}^T \underline{K} \underline{U} - \underline{U}^T \underline{R} \tag{9}$$

where $\underline{k}$ is the stiffness matrix for the complete system and is given by the sum of the element stiffnesses

$$\underline{K} = \sum_{m=1}^{M} \underset{-}{K}_m \tag{10}$$

$$\underset{-}{K}_m = \int_{Vol} \underset{-}{a}_m^T \underset{-}{c}_m \underset{-}{a}_m \, dV_m \tag{11}$$

The generalized forces matrix $\underline{R}$ is defined as

$$R = \sum_{m=1}^{M} \int_{Vol} \underset{-}{d}_m^T \underset{-}{F}_m \, dV_m + \sum_{m=1}^{M} \int_{Area} \underset{-}{d}_m^T \underset{-}{T}_m \, dA_m \tag{12}$$

The potential energy is minimized by requiring that

$$\frac{\partial \Phi}{\partial U_i} = 0 \qquad i = 1, 2 \ldots., I \tag{13}$$

where I is the total number of unknown generalized coordinates. The application of Eq. (13) yields the following set of linear equations which govern the equilibrium of a finite element system.

$$\underline{K}\ \underline{U} = \underline{R} \tag{14}$$

For the prismatic space formulation given in this paper the displacement field is composed or orthogonal functions and it may be shown that Eq. (14) is of the following uncoupled form:

$$\begin{bmatrix} \underline{K}^{(0)} & & & & & \\ & \underline{K}^{(1)} & & & & \\ & & \underline{K}^{(3)} & & & \\ & & & \cdot & & \\ & & & & \cdot & \\ & & & & & \cdot \end{bmatrix} \begin{bmatrix} \underline{U}^{(0)} \\ \underline{U}^{(1)} \\ \underline{U}^{(2)} \\ \cdot \\ \cdot \\ \cdot \end{bmatrix} \begin{bmatrix} \underline{R}^{(0)} \\ \underline{R}^{(1)} \\ \underline{R}^{(2)} \\ \cdot \\ \cdot \\ \cdot \end{bmatrix} \tag{15}$$

Or the typical uncoupled equation for the n^{th} harmonic function is given by

$$\underline{K}^{(n)}\ \underline{U}^{(n)} = \underline{R}^{(n)} \tag{16}$$

Therefore, the generalized coordinates are found by solving several small sets of linear equations rather than solving one large set of equations.

B. Displacement Field Approximation

As shown in Figure 2, the geometry of a typical prismatic space element is defined by three points in the x-y plane i, j and k and the common element length in the z-direction ℓ. The only assumption



FIGURE 2 THE PRISMATIC SPACE FINITE ELEMENT

required in a finite element analysis is on the form of the displacement field within each element. For a typical element the three components of displacement are approximated by

$$u_x\ (x,y,z) = \sum_{n=0}^{N} u_x^{(n)}\ (x,y)\ \text{Cos}\ \eta z$$

$$u_y\ (x,y,z) = \sum_{n=0}^{N} u_y^{(n)}\ (x,y)\ \text{Cos}\ \eta z \tag{17}$$

$$u_z\ (x,y,z) = \sum_{n=1}^{N} u_z^{(n)}\ (x,y)\ \text{Sin}\ \eta z$$

$$\text{where}\quad \eta = \frac{n\pi}{\ell}$$

The number of terms N required will depend on the specific problem considered and, generally, is governed by the accuracy in which the load can be expressed in a trigonometric series. It is of interest to note that for $n=0$ the displacement field is equivalent to the plane strain approximation. Within each element m the two-dimensional approximate displacements functions are given by

$$u_{xm}\ (x,y) = \alpha_{1m}^{(n)} + \alpha_{2m}^{(n)}\ x + \alpha_{3m}^{(n)}\ y$$

$$u_{ym}\ (x,y) = \beta_{1m}^{(n)} + \beta_{2m}^{(n)}\ x + \beta_{3m}^{(n)}\ y \tag{18}$$

$$u_{zm}\ (x,y) = \gamma_{1m}^{(n)} + \gamma_{2m}^{(n)}\ x + \gamma_{3m}^{(n)}\ y$$

Therefore, the three-dimensional displacement field approximation involves linear functions in the x-y plane multiplied by trigonometric functions in the z-direction.

The constants α, β and γ may be expressed in terms of the generalized displacements at the connecting nodal points i, j and k. Or

$$\begin{bmatrix} \alpha_{1m}^{(n)} & \beta_{1m}^{(n)} & \gamma_{1m}^{(n)} \\ \alpha_{2m}^{(n)} & \beta_{2m}^{(n)} & \gamma_{2m}^{(n)} \\ \alpha_{3m}^{(n)} & \beta_{3m}^{(n)} & \gamma_{3m}^{(n)} \end{bmatrix} = \underset{\sim}{D}_m \begin{bmatrix} U_{xi}^{(n)} & U_{yi}^{(n)} & U_{zi}^{(n)} \\ U_{xj}^{(n)} & U_{yj}^{(n)} & U_{zj}^{(n)} \\ U_{xk}^{(n)} & U_{yk}^{(n)} & U_{zk}^{(n)} \end{bmatrix}$$

where

$$D_m = \frac{1}{\lambda} \begin{bmatrix} x_j\, y_k - x_k\, y_j & x_k\, y_i - x_i\, y_k & x_2\, y_2 - x_j\, y_i \\ y_j - y_k & y_k - y_i & y_i - y_j \\ x_k - x_j & x_i - x_k & x_j - x_i \end{bmatrix}$$

where $\lambda = x_j\,(y_k - y_i) + x_i\,(y_j - y_k) + x_k\,(y_i - y_j)$

The total strains within an element are expressed in terms of a summation of the strains which are associated with each harmonic. The strains associated with each harmonic are obtained by differentiation of the assumed displacement field, Equation (17), and are given by

$$\varepsilon_{xx}^{(n)} = \alpha_2^{(n)} \text{ Cos } \eta z$$

$$\varepsilon_{yy}^{(n)} = \beta_3^{(n)} \text{ Cos } \eta z$$

$$\varepsilon_{zz}^{(n)} = \eta\ (\gamma_1^{(n)} + \gamma_2^{(n)}\ x + \gamma_3^{(n)}\ y)\ \mathrm{Cos}\ \eta z$$

$$\varepsilon_{xy}^{(n)} = (\alpha_3^{(n)} + \beta_2^{(n)})\ \mathrm{Cos}\ \eta z$$

$$\varepsilon_{xz}^{(n)} = [-\ \eta\ (\alpha_1^{(n)} + \alpha_2^{(n)}\ x + \alpha_3^{(n)}\ y) + \gamma_2^{(n)}]\ \mathrm{Sin}\ \eta z \qquad (20)$$

$$\varepsilon_{yz}^{(n)} = [-\ \eta\ (\beta_1^{(n)} + \beta_2^{(n)}\ x + \beta_3^{(n)}\ y) + \gamma_3^{(n)}]\ \mathrm{Sin}\ \eta z$$

C. Harmonic Element Stiffness

If the constants α, β and γ are eliminated from Equation (20) by a substitution of Equation (19) the following expression for the harmonic strains in terms of the generalized displacements is developed:

$$\underline{\varepsilon}_m^{(n)} = \underline{g}^{(n)}\ \underline{h}_m^{(n)} \begin{bmatrix} \underline{D}_m & 0 & 0 \\ 0 & \underline{D}_m & 0 \\ 0 & 0 & \underline{D}_m \end{bmatrix} \begin{bmatrix} U_x^{(n)} \\ U_y^{(n)} \\ U_z^{(n)} \end{bmatrix} \qquad (21)$$

Or in terms of the notation of Equation (6)

$$\underline{\varepsilon}_m^{(n)} = \underline{a}_m^{(n)}\ \underline{U}^{(n)} \qquad (22)$$

in which

$$
\underline{g}^{(n)} = \begin{bmatrix} \cos \eta z & 0 & 0 & 0 & 0 & 0 \\ 0 & \cos \eta z & 0 & 0 & 0 & 0 \\ 0 & 0 & \cos \eta z & 0 & 0 & 0 \\ 0 & 0 & 0 & \cos \eta z & 0 & 0 \\ 0 & 0 & 0 & 0 & \sin \eta z & 0 \\ 0 & 0 & 0 & 0 & 0 & \sin \eta z \end{bmatrix} \quad (23)
$$

$$
\underline{h}^{(n)} = \begin{bmatrix} 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \eta & \eta x & \eta y \\ 0 & 0 & 1 & 0 & 1 & 0 & 0 & 0 & 0 \\ -\eta & -\eta x & -\eta y & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -\eta & -\eta x & -\eta y & 0 & 0 & 1 \end{bmatrix} \quad (24)
$$

From Equation (11) the element stiffness matrix associated with harmonic "n" is given by

$$
\underline{k}_m^{(n)} = \int_{Vol} \underline{a}_m^{(n)^T} \underline{c}_m \underline{a}_m^{(n)} \, dx \, dy \, dz \quad (25)
$$

The substitution of Equation (22) into Equation (25) and the integration with respect to z from $-\ell$ to ℓ yields

$$
\underline{k}_m^{(n)} = \delta^{(n)} \begin{bmatrix} \underline{D}^T & . & . \\ . & \underline{D}^T & . \\ . & . & \underline{D}^T \end{bmatrix} \int_{Area} \underline{h}_m^{(n)} \underline{c}_m \underline{h}_m^{(n)} \, dx \, dy \begin{bmatrix} \underline{D} & . & . \\ . & \underline{D} & . \\ . & . & \underline{D} \end{bmatrix} \quad (26)
$$

where

$$\delta^{(n)} = 2\,\ell \quad \text{for} \quad n = 0$$

$$\delta^{(n)} \quad \ell \quad \text{for} \quad n > 0$$

For an isotropic material the matrix to be integrated is given on the following page.

where

$$\varepsilon = \frac{E\,(1-\nu)}{(1-2\,\nu)\,(1+\nu)}$$

$$\mu = \frac{E\,\nu}{(1-2\,\nu)\,(1+\nu)}$$

E = Modulus of Elasticity

ν = Poisson's Ratio

The integrals required in Equation (25) are easily evaluated in closed form for a triangular area.

D. Harmonic Loads

The determination of the generalized forces which are associated with each harmonic may be obtained by the evaluation of Equation (12). Because of the arbitrary nature of the loading, the details of this development will not be given, However, it is of interest to point out that these generalized forces are essentially the same as the coefficient of the expansion of the loading in a Fourier Series.

$$\underline{h}\,(x,y)^T\;\underline{C}\;\underline{h}\,(x,y) =$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \varepsilon & 0 & 0 & 0 & \mu & \eta\mu & \eta\mu x & \eta\mu y \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \mu & 0 & 0 & 0 & \varepsilon & \eta\mu & \eta\mu x & \eta\mu y \\ 0 & \eta\mu & 0 & 0 & 0 & \eta\mu & \eta^2\varepsilon & \eta^2 x\varepsilon & \eta^2 y\varepsilon \\ 0 & \eta\mu x & 0 & 0 & 0 & \eta\mu x & \eta^2 x\varepsilon & \eta^2 x^2\varepsilon & \eta^2 xy\varepsilon \\ 0 & \eta\mu y & 0 & 0 & 0 & \eta\mu y & \eta^2 y\varepsilon & \eta^2 xy\varepsilon & \eta^2 y^2\varepsilon \end{bmatrix} +$$

$$G \begin{bmatrix} \eta^2 & \eta^2 x & \eta^2 y & 0 & 0 & 0 & 0 & -\eta & 0 \\ \eta^2 x & \eta^2 x^2 & \eta^2 xy & 0 & 0 & 0 & 0 & -\eta x & 0 \\ \eta^2 y & \eta^2 xy & 1+\eta^2 y^2 & 0 & 1 & 0 & 0 & -\eta y & 0 \\ 0 & 0 & 0 & \eta^2 & \eta^2 x & \eta^2 y & 0 & 0 & -\eta \\ 0 & 0 & 1 & \eta^2 x & 1+\eta^2 x^2 & \eta^2 xy & 0 & 0 & -\eta x \\ 0 & 0 & 0 & \eta^2 y & \eta^2 xy & \eta^2 y^2 & 0 & 0 & -\eta y \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -\eta & -\eta x & -\eta y & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -\eta & -\eta x & -\eta y & 0 & 0 & 0 \end{bmatrix} \quad (27)$$

EXAMPLES

The accuracy of the prismatic space finite element approach is checked by comparing it with the classical Baussinesg solutions for surface loads on an elastic half space. The two solutions considered here are:

1. Point load
2. Uniform pressure on a rectangular area

The finite element mesh used for both solutions is shown in Figure 12. A period length of 200 inches is assumed while the elastic coefficients used are indicated in Figure 12.

The point load is approximated by 19 harmonics. A comparison of vertical stress distribution with depth is shown in Figure 3a. The finite element solution differs from the Baussinesg solution only at points near the load. This is an area of high strain rates which affect the accuracy of the finite element approach.

Figure 3b indicates that the surface deflection is underestimated by using finite elements. Fixed boundaries at a finite depth are used in the finite element idealization, while the Baussinesg solution is for a depth of infinite extent. Surface deflections would, therefore, always be underestimated by finite elements (when compared with Baussinesg solutions) unless the depth of the boundary is substantially increased.

A dimensionless plot of vertical stress distribution with depth for a uniform pressure over a rectangular area is shown in Figure 3c. The load is approximated by 9 harmonics. The prismatic space finite element solution again compares very favorably with the elastic solution, except at points very close to the load.



FIG. 3a VERTICAL STRESS DISTRIBUTION WITH DEPTH FOR A POINT LOAD ON A SEMI INFINITE MASS



FIG. 3b SURFACE DEFLECTION DUE TO A POINT LOAD ON A SEMI INFINITE MASS



FIG.3c VERTICAL STRESS DISTRIBUTION WITH DEPTH FOR A UNIFORM LOAD ON A SQUARE AREA

In Figure 4, the contribution of each harmonic to the total maximum deflection is indicated for each of the examples considered. A total number of 12 or 13 harmonics for the point load would have sufficed. Because a complete finite element solution is obtained for each harmonic, the program is very time consuming. It is therefore essential to limit the number of harmonics to a minimum and still maintain accuracy.

Another appropriate check can be made by comparing an "axisymmetric" prismatic space solution with a solution obtained from the axisymmetric finite element approach. Both there programs find wide application in the field of highway pavement research at the moment. A pavement consisting of a concrete (or soil-cement) base on a clay subgrade is therefore selected for this purpose. The finite element mesh representing the pavement is shown in Figure 13. Also indicated are Young's module, Poisson's ration and the load. In the prismatic space program, the load is approximated as a time pressure of 25 psi applied over a 16" square area. The number of harmonics used is 16. A large deflection basin is to be expected for the stiff base under consideration, and a period length of 240 inches is used to isolate the influence of the periodically applied load.

A comparison of the surface deflection profile is shown in Figure 5. Very good agreement is achieved except that the prismatic space solutions for the x-y and y-z planes differ slightly for points further away from the load. This difference is due to the Fourier expansion of the load in the y-z plane.

A comparison of vertical stress distributions with depth for a vertical line slightly offset from the axis of symmetry is shown in Figure 6. A slight difference in radial distribution of vertical stress



FIG. 4 CONTRIBUTION OF EACH HARMONIC TO THE TOTAL MAXIMUM DEFLECTION



FIG. 5 SURFACE DEFLECTION PROFILE OF A CONCRETE PAVEMENT ON A CLAY SUBGRADE



FIG. 6 VERTICAL STRESS DISTRIBUTION WITH DEPTH FOR 6400 LB. WHEEL LOAD ON PAVEMENT

exists near the surface due to the difference in shape of the loaded areas (see Fig. 7). This accounts for the difference in vertical stress near the surface as shown in Figure 6.

It can be concluded from the foregoing examples that the prismatic space approach yields results of acceptable accuracy, provided the necessary precautions regarding number of harmonics, period length, etc., are taken.

An entire new field of problems can now be analyzed that were previously unsolved or had solutions for limited boundary conditions only.

One such problem is the case of a point load on a quarter space. This problem is analyzed using the same mesh as shown in Figure 12, with the nodal point restraints in the y-z plane at x=0 removed. Nineteen harmonics and a period length of 200 inches are used again.

Nodal point displacements in the x-y and y-z planes are shown in Figures 8 and 9 respectively. Horizontal and vertical stress contours are plotted in Figures 10 and 11.

As an example of the practical application of the prismatic space program, reference is made to Figure 14. The surface displacement of a pavement loaded near a vertical joint is shown. The program as used here shows the high tensile stresses in the upper fibres of the layers to the left of the joint are to be expected. The majority of the displacement occurs in the soft subgrade. Because continuity between layers is assumed in the finite element idealization, the layers to the left of the joint are forced down with the subgrade, resulting in a high curvature and stresses. In practice, the band between the clay subgrade



FIG. 7 VERTICAL STRESS DISTRIBUTION ON PLANE 2" BELOW SURFACE OF PAVEMENT



FIG. 8 GRID DEFORMATION IN X-Y PLANE FOR A POINT LOAD ON QUARTER SPACE



FIG. 9 GRID DEFORMATION IN Y-Z PLANE FOR A POINT LOAD ON QUARTER SPACE



FIG. 10 HORIZONTAL STRESS CONTOURS (x 10^{-5} PSI) FOR A POINT LOAD ON QUARTER SPACE



FIG. 11 VERTICAL STRESS CONTOURS (x 10^{-5} PSI)
FOR A POINT LOAD ON QUARTER SPACE



FIG. 12 FINITE ELEMENT MESH FOR POINT AND RECTANGULAR LOADS IN HALF SPACE AND A POINT LOAD IN QUARTER SPACE (WITH NODAL POINT RESTRAINTS IN Y-Z PLANE AT X=0 REMOVED)



FIG. 13 FINITE ELEMENT MESH OF HIGHWAY PAVEMENT USED FOR COMPARISON OF PRISMATIC SPACE AND AXISYMMETRIC PROGRAMS.



FIG. 14 SURFACE DISPLACEMENT PROFILE FOR A LOAD CLOSE TO A VERTICAL CONSTRUCTION JOINT

and the pavement will be too low to maintain continuity. A more realistic prediction of stress distribution will be obtained by reducing the stiffness of the pavement to the left of the joint to a small valve compared with the pavement stiffness.

COMPUTER PROGRAM INPUT

The first step is to select a finite element representation of the two-dimensional cross-section of the body. Elements and nodal points are then numbered in two numerical sequences, each starting with one. The following group of punched cards numerically defines the two-dimensional structure to be analyzed.

A. Identification Card - (12A6)

Columns	1 - 72	Of this card contain information to be with results.

B. Control Card - (4I5, F10.2)

Columns	1 - 5	Number of nodal points (350 max.)
	6 - 10	Number of elements (300 max.)
	11 - 15	Number of different materials (12 max.)
	16 - 20	Number of harmonics
	21 - 30	ZL - length

The ZL length is half of the period length (length to next load).

C. Material Property Information

The following group of cards for each material:

First Card: (2I5, 2F10.0)

Columns	1 - 5	Material identification - any number from 1 - 12
	6 - 10	Number of different Y-ordinates for which properties are given - 8 maximum

Second Card: (3F10.0) One card for each Y-ordinate

Columns	1 - 10	Y-ordinate of material for which property is given
	11 - 20	Young's Modulus
	21 - 30	Poisson's Ratio

D. Nodal Point Data - (I5, F5.0, 5F10.0)

One card for each nodal point with the following information:

Columns	1 - 5	Nodal point number
	6 - 10	Code number
	11 - 20	X-ordinate
	21 - 30	Y-ordinate
	31 - 40	UX
	41 - 50	UY
	51 - 60	Loaded length in Z-direction

If the number of column 10 is

0, UX is specified X-load
UY is specified Y-load

1, UX is specified X-displacement
UY is specified Y-load

2, UX is specified X-load
UY is specified Y-displacement

3, UX is specified X-displacement
UY is specified Y-displacement

All loads are total forces acting on the nodal point. Loaded length is half the total loaded length in the Z-direction. Nodal point cards must be in numerical sequence. If cards are omitted, the omitted nodal points are generated at equal intervals along a straight line between the defined nodal points. The boundary code (column 10), UX, UY, and loaded length, are set equal to zero.

E. Element Properties: (6I5)

One card for each element

Columns	1 - 5	Element number
	6 - 10	Nodal point I
	11 - 15	Nodal point J
	16 - 20	Nodal point K
	21 - 25	Nodal point L
	26 - 30	Material Identification

F. Longitudinal Distance: (F10.0)

Column	1 - 10	ZZ - length

This distance defines the cross-sections in the longitudinal (Z) direction where stresses and strains are to be printed. Nodal point displacements and element stresses for the defined finite element mesh are obtained.

REFERENCE

1. Clough, R. W., "The Finite Element Method in Plane Stress Analysis," Proc. Am. Soc. Civil Engrs., 345 - 378 (1960).

2. Clough, R.W., and Tocher, J. L., "Finite Element Stiffness Matrices for the Analysis of Plate Bending," Proc. 1st Conf. on Matrix Methods, Wright-Patterson AEB, Ohio, (1965)

3. Clough, R. W. and Rashid, Y. R. "Finite Element Analysis of Axisymmetric Solids," Proc. ASCE, Eng. Mech. Div. 91 (1965) 71.

4. Wilson, E. L., Structural Analysis of Axisymmetric Solids, AIAA J. 3 (1965) 2269.

5. Ahmad, S., Irons, B. and Zienkiewicz, O., "Analysis of Thick and Thin Shell Structures by Curved Finite Elements," International Journal for Numerical Methods in Engineering, Vol. 2, No. 3, July 1970, pp. 419 - 451.

6. Pawsey, S., "Analysis of Moderately Thick to Thin Shells by Finite Elements," Ph.D. Thesis, University of California, Berkeley, September 1970.

7. Zienkiewicz, O. C., and Cheung, Y. K., "The Finite Element Method in Structural and Continuum Mechanics," McGraw-Hill, 1967.

8. Herrmann, L. R., "Elastic Torsional Analysis of Irregular Shapes," ASCE EM6 December 1965.

9. Wilson, E. L., and Nickell, R. E., "Application of the Finite Element Method to Heat Condition Analysis," Nuclear Engineering and Design 4 (1966) 276-886

10. Sandhu, R. S., and Wilson, E. L., "Finite Element Analysis of Seepage in Elastic Media," ASCE EM3 (1969) 641-652.

FORTRAN IV COMPUTER PROGRAM LISTING

```
      PROGRAM MAIN(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT,TAPE1,TAPE2,
     1 TAPE4)
C     ANALYSIS OF ARBITRARY PRISMATIC SOLIDS
C
      COMMON NUMNP,NUMEL,NUMMAT,NUMHAR,XHAR,ZL,PI,HED(12),C11,C12,G
     1 ,MBAND,NUMBLK,B(108),A(108,54)
      COMMON /ELEARG/ IX(300,5),E(8,3,12),NUMTC(12),T(400),SIG(300,6)
      COMMON /ORDARG/ X(350),Y(350),UX(350),UY(350),UZ(350),CODE(350)
C
      REWIND 4
      BAD=0.0
      PI=3.1415927
C*********************************************************************
C     READ AND PRINT OF CONTROL INFORMATION AND MATERIAL PROPERTIES
C*********************************************************************
   50 READ  (5,1000) HED,NUMNP,NUMEL,NUMMAT,NUMHAR,ZL
      WRITE (6,2000) HED,NUMNP,NUMEL,NUMMAT,NUMHAR,ZL
C
   56 DO 59 M=1,NUMMAT
      READ (5,1001) MTYPE,NUMTC(MTYPE)
      WRITE (6,2011) MTYPE,NUMTC(MTYPE)
      NUM=NUMTC(MTYPE)
      READ (5,1005) ((E(I,J,MTYPE),J=1,3),I=1,NUM)
      WRITE (6,2010) ((E(I,J,MTYPE),J=1,3),I=1,NUM)
   59 CONTINUE
C
C*********************************************************************
C     READ AND PRINT OF NODAL POINT DATA
C*********************************************************************
      WRITE (6,2004)
      L=0
   60 READ  (5,1002) N,CODE(N),X(N),Y(N),UX(N),UY(N),T(N)
      NL=L+1
      ZX=N-L
      DX=(X(N)-X(L))/ZX
      DY=(Y(N)-Y(L))/ZX
   70 L=L+1
      IF(N-L) 100,90,80
   80 CODE(L)=0.0
      X(L)=X(L-1)+DX
      Y(L)=Y(L-1)+DY
      UX(L)=0.0
      UY(L)=0.0
       T(L)=0.0
      GO TO 70
   90 WRITE (6,2002) (K,CODE(K),X(K),Y(K),UX(K),UY(K),T(K),K=NL,N)
      IF(NUMNP-N) 100,110,60
  100 WRITE (6,2009) N
      CALL EXIT
  110 CONTINUE
```

```
C*********************************************************************
C        READ AND PRINT OF ELEMENT PROPERTIES
C*********************************************************************
      WRITE (6,2001)
      N=0
  130 READ (5,1003) M,(IX(M,I),I=1,5)
  140 N=N+1
      IF (M-N) 170,170,150
  150 IX(N,1)=IX(N-1,1)+1
      IX(N,2)=IX(N-1,2)+1
      IX(N,3)=IX(N-1,3)+1
      IX(N,4)=IX(N-1,4)+1
      IX(N,5)=IX(N-1,5)
  170 WRITE (6,2003) N,(IX(N,I),I=1,5)
      IF (M-N) 180,180,140
  180 IF (NUMEL-N) 190,190,130
  190 CONTINUE
C*********************************************************************
C     DETERMINE BAND WIDTH
C*********************************************************************
      J=0
      DO 340 N=1,NUMEL
      DO 340 I=1,4
      DO 325 L=1,4
      KK=IABS(IX(N,I)-IX(N,L))
      IF(18-KK) 310,310,315
  310 WRITE (6,2008) N
      BAD=1.0
  315 IF (KK-J) 325,325,320
  320 J=KK
  325 CONTINUE
  340 CONTINUE
      MBAND=3*J+3
C*********************************************************************
C     DETERMINE DISPLACEMENTS AND STRESSES FOR EACH HARMONIC
C*********************************************************************
C
      IF (BAD) 350,345,350
  350 CALL EXIT
  345 NUMUK=3*NUMNP
      DO 500 NHAR=1,NUMHAR
      XHAR=NHAR-1
C
C     FORM STIFFNESS MATRIX
C
      CALL STIFF
C
C     SOLVE FOR HARMONIC DISPLACEMENTS
C
      CALL BANSOL
C
      WRITE (6,2006) (N,B(3*N-2),B(3*N-1),B(3*N),N=1,NUMNP)
      WRITE (4) (B(N),N=1,NUMUK)
C
      CALL STRESS
C
  500 CONTINUE
```

```
C***********************************************************************
      CALL OUTPUT
C***********************************************************************
      STOP
 1000 FORMAT (12A6/4I5,1F10.2)
 1001 FORMAT (2I5,2F10.0)
 1002 FORMAT (I5,F5.0,5F10.0)
 1003 FORMAT (6I5)
 1005 FORMAT (3F10.0)
 2000 FORMAT (1H1 12A6/
     1 30H0 NUMBER OF NODAL POINTS------ I3 /
     2 30H0 NUMBER OF ELEMENTS---------- I3 /
     3 30H0 NUMBER OF DIFF. MATERIALS--- I3 /
     4 30H0 NUMBER OF HARMONICS--------- I3 /
     5 30H0 Z-LENGTH-------------------- F10.3)
 2001 FORMAT (49H1ELEMENT NO.        I      J      K      L    MATERIAL  )
 2003 FORMAT (1I13,4I6,1I12)
 2004 FORMAT  (48H1NODAL POINT        TYPE  X-ORDINATE  Y-ORDINATE
     118X 6HX-LOAD 18X 6HY-LOAD 2X 12HLOAD LENGTH )
 2002 FORMAT (I12,F12.2,2F12.3,2E24.7,F12.3)
 2006 FORMAT (12H1N.P. NUMBER 18X 2HUX 18X 2HUY 18X 2HUZ /(1I12,3E20.7))
 2007 FORMAT (2I6,F12.3)
 2008 FORMAT (30H0N.P. DIFF. TOO LARGE EL.NO.= I4)
 2009 FORMAT (26H0NODAL POINT CARD ERROR N= I5)
 2010 FORMAT ( 14X 1HY 14X 1HE 13X 2HNU / (F15.3,E15.7,F15.3))
 2011 FORMAT (17H0MATERIAL NUMBER= I3, 20H, NUMBER OF Y CARDS=  I3)
C
      END
```

```
      SUBROUTINE STIFF
C
      COMMON NUMNP,NUMEL,NUMMAT,NUMHAR,XHAR,ZL,PI,HED(12),C11,C12,G
     1 ,MBAND,NUMBLK,B(108),A(108,54)
      COMMON /ELEARG/ IX(300,5),E(8,3,12),NUMTC(12),T(400),SIG(300,6)
      COMMON /ORDARG/ X(350),Y(350),UX(350),UY(350),UZ(350),CODE(350)
      COMMON /QUADAR/ N,VOL,MTYPE,ST(15,15),DD(3,3),D(3,3),S(9,9),
     1 XX(3),YY(3),II,JJ,EE(7),LM(4),DT(3,5),U(15),EPS(6)
C
C***********************************************************************
C     INITIALIZATION
C***********************************************************************
      REWIND 2
      NB=18
      ND=3*NB
      ND2=2*ND
      NUMBLK=0
      DO 40 N=1,NUMEL
   40 IX(N,5)=IABS(IX(N,5))
C
      DO 50 N=1,ND2
      B(N)=0.0
      DO 50 M=1,ND
   50 A(N,M)=0.0
C***********************************************************************
C     FORM STIFFNESS MATRIX IN BLOCKS
C***********************************************************************
   60 NUMBLK=NUMBLK+1
      NH=NB*(NUMBLK+1)
      NM=NH-NB
      NL=NM-NB+1
      KSHIFT=3*NL-3
C
      DO 210 N=1,NUMEL
C
      IF (IX(N,5)) 210,210,65
   65 DO 80 I=1,4
      IF (IX(N,I)-NL) 80,70,70
   70 IF (IX(N,I)-NM) 90,90,80
   80 CONTINUE
      GO TO 210
C
   90 CALL QUAD
      IX(N,5)=-MTYPE
C
C             ADD ELEMENT STIFFNESS TO TOTAL STIFFNESS
C
  165 DO 166 I=1,4
  166 LM(I)=3*IX(N,I)-3
C
      DO 200 I=1,4
      DO 200 K=1,3
      II=LM(I)+K-KSHIFT
      KK=3*I-3+K
      DO 200 J=1,4
      DO 200 L=1,3
```

```
      JJ=LM(J)+L-II+1-KSHIFT
      LL=3*J-3+L
      IF(JJ) 200,200,195
  195 A(II,JJ)=A(II,JJ)+ST(KK,LL)
  200 CONTINUE
  210 CONTINUE
C
C     ADD CONCENTRATED FORCES WITHIN BLOCK
C
      DO 250 N=NL,NM
      IF (N-NUMNP) 240,240,310
  240 K=3*N-KSHIFT-1
      IF (T(N).EQ.0.0) TM=1.0
      IF (XHAR.EQ.0.0) TM=0.5
      IF (T(N)*XHAR.NE.0.0) TM=ZL*SIN(XHAR*PI*T(N)/ZL)/(XHAR*PI*T(N))
  245 B(K)=B(K)+UY(N)*TM
  250 B(K-1)=B(K-1)+UX(N)*TM
C
C     BOUNDARY CONDITIONS
C
  310 DO 400 M=NL,NH
      IF (M-NUMNP) 315,315,400
  315 U=UX(M)
      N=3*M-1-KSHIFT-1
      IF (CODE(M)) 390,400,316
  316 IF (CODE(M)-1.) 317,370,317
  317 IF (CODE(M)-2.) 318,390,318
  318 IF (CODE(M)-3.) 390,380,390
  370 CALL MODIFY(A,B,ND2,MBAND,N,U)
      GO TO 400
  380 CALL MODIFY(A,B,ND2,MBAND,N,U)
  390 U=UY(M)
      N=N+1
      CALL MODIFY(A,B,ND2,MBAND,N,U)
  400 CONTINUE
C
C     WRITE BLOCK OF EQUATIONS ON TAPE AND SHIFT UP LOWER BLOCK
C
      WRITE (2) (B(N),(A(N,M),M=1,MBAND),N=1,ND)
C
      DO 420 N=1,ND
      K=N+ND
      B(N)=B(K)
      B(K)=0.0
      DO 420 M=1,ND
      A(N,M)=A(K,M)
  420 A(K,M)=0.0
C
C     CHECK FOR LAST BLOCK
C
      IF (NM-NUMNP) 60,480,480
  480 CONTINUE
C**************************************************************************
  500 RETURN
C
      END
```

```
      SUBROUTINE TRISTF
C
      COMMON NUMNP,NUMEL,NUMMAT,NUMHAR,XHAR,ZL,PI,HED(12),C11,C12,G
     1 ,MBAND,NUMBLK,B(108),A(108,54)
      COMMON /ELEARG/ IX(300,5),E(3,3,12),NUMTC(12),T(400),SIG(300,6)
      COMMON /ORDARG/ X(350),Y(350),UX(350),UY(350),UZ(350),CODE(350)
      COMMON /QUADAR/ N,VOL,MTYPE,ST(15,15),DD(3,3),D(3,3),S(9,9),
     1 XX(3),YY(3),II,JJ,EE(7),LM(4),DT(3,5),U(15),EPS(6)
C
C        1. INITIALIZATION
C
      XX(1)=X(II)
      XX(2)=X(JJ)
      YY(1)=Y(II)
      YY(2)=Y(JJ)
C
      COMM=XX(2)*(YY(3)-YY(1))+XX(1)*(YY(2)-YY(3))+XX(3)*(YY(1)-YY(2))
      AREA=COMM/2.0
      V=AREA*ZL
      G=0.5*EE(1)/(1.+EE(2))
      C11=EE(1)/((1.+EE(2))*(1.-2.*EE(2)))
      C12=EE(2)*C11
      C11=C11-C12
      C13=C12
      C22=C11
      C23=C12
      C33=C11
      COM=PI*XHAR/ZL
      COM2=COM**2
C
      SUMX=XX(1)+XX(2)+XX(3)
      SUMY=YY(1)+YY(2)+YY(3)
      XI=SUMX/3.
      YI=SUMY/3.
      X2I=(XX(1)*(SUMX+XX(1))+XX(2)*(SUMX+XX(2))+XX(3)*(SUMX+XX(3)))/12.
      XYI=(YY(1)*(SUMX+XX(1))+YY(2)*(SUMX+XX(2))+YY(3)*(SUMX+XX(3)))/12.
      Y2I=(YY(1)*(SUMY+YY(1))+YY(2)*(SUMY+YY(2))+YY(3)*(SUMY+YY(3)))/12.
C
      DO 100 I=1,81
  100 S(I,1)=0.0
C
C        2. FORM   COEFFICIENT-DISPLACEMENT TRANSFORMATION MATRIX
C
      DD(1,1)=(XX(2)*YY(3)-XX(3)*YY(2))/COMM
      DD(1,2)=(XX(3)*YY(1)-XX(1)*YY(3))/COMM
      DD(1,3)=(XX(1)*YY(2)-XX(2)*YY(1))/COMM
      DD(2,1)=(YY(2)-YY(3))/COMM
      DD(2,2)=(YY(3)-YY(1))/COMM
      DD(2,3)=(YY(1)-YY(2))/COMM
      DD(3,1)=(XX(3)-XX(2))/COMM
      DD(3,2)=(XX(1)-XX(3))/COMM
      DD(3,3)=(XX(2)-XX(1))/COMM
```

```
C
C        3. FORM INTEGRAL(G)T*(C)*(G)
C
      TM=G*V*COM2
      S(1,1)=TM
      S(1,2)=TM*XI
      S(1,3)=TM*YI
      S(2,2)=TM*X2I
      S(2,3)=TM*XYI
      S(3,3)=TM*Y2I
C
      TM=-COM*G*V
      S(1,8)=TM
      S(2,8)=TM*XI
      S(3,8)=TM*YI
C
      TM=C33/G
      DO 150 I=1,3
      S(I+3,9)=S(I,8)
      DO 150 J=1,3
      S(I+3,J+3)=S(I,J)
  150 S(I+6,J+6)=S(I,J)*TM
      TM=G*V
      S(3,3)=S(3,3)+TM
      S(5,5)=S(5,5)+TM
      S(3,5)=TM
      S(8,8)=S(8,8)+TM
      S(9,9)=S(9,9)+TM
C
      TM=COM*C13*V
      S(2,7)=TM
      S(2,8)=S(2,8)+TM*XI
      S(2,9)=TM*YI
C
      TM=COM*C23*V
      S(6,7)=TM
      S(6,8)=TM*XI
      S(6,9)=S(6,9)+TM*YI
C
      S(2,2)=S(2,2)+C11*V
      S(2,6)=C12*V
      S(6,6)=S(6,6)+C22*V
C
      DO 200 I=1,9
      DO 200 J=I,9
  200 S(J,I)=S(I,J)
C
```

```
C         5. FORM ELEMENT STIFFNESS MATRIX (H)T*(D)*(H)
C
      DO 500 I=1,3
      DO 500 J=1,3
C
      DO 300 K=1,3
      DO 300 L=1,3
      D(K,L)=0.0
      DO 300 M=1,3
      MM=3*J-3+M
      KK=3*I-3+K
  300 D(K,L)=D(K,L)+S(KK,MM)*DD(M,L)
C
      DO 400 K=1,3
      DO 400 L=1,3
      KK=3*I-3+K
      LL=3*J-3+L
      S(KK,LL)=0.0
      DO 400 M=1,3
  400 S(KK,LL)=S(KK,LL)+DD(M,K)*D(M,L)
C
  500 CONTINUE
C
  700 RETURN
C
      END



      SUBROUTINE MODIFY(A,B,NEQ,MBAND,N,U)
C
      DIMENSION A(108,54),B(108)
C
      DO 250 M=2,MBAND
      K=N-M+1
      IF(K) 235,235,230
  230 B(K)=B(K)-A(K,M)*U
      A(K,M)=0.0
  235 K=N+M-1
      IF(NEQ-K) 250,240,240
  240 B(K)=B(K)-A(N,M)*U
      A(N,M)=0.0
  250 CONTINUE
      A(N,1)=1.0
      B(N)=U
      RETURN
C
      END
```

```
      SUBROUTINE QUAD
C
      COMMON NUMNP,NUMEL,NUMMAT,NUMHAR,XHAR,ZL,PI,HED(12),C11,C12,G
     1 ,MBAND,NUMBLK,B(108),A(108,54)
      COMMON /ELEARG/ IX(300,5),E(8,3,12),NUMTC(12),T(400),SIG(300,6)
      COMMON /ORDARG/ X(350),Y(350),UX(350),UY(350),UZ(350),CODE(350)
      COMMON /QUADAR/ N,VOL,MTYPE,ST(15,15),DD(3,3),D(3,3),S(9,9),
     1 XX(3),YY(3),II,JJ,EE(7),LM(4),DT(3,5),U(15),EPS(6)
C
      DO 50 I=1,15
   50 DT(I,1)=0.0
C
C     SELECT MATERIAL PROPERTIES
C
      I=IX(N,1)
      J=IX(N,2)
      K=IX(N,3)
      L=IX(N,4)
      MTYPE=IABS(IX(N,5)))
C
      DO 105 KK=1,2
  105 EE(KK)=E(1,KK+1,MTYPE)
C
C     CALCULATE CENTRAL POINT
      XX(3)=(X(I)+X(J)+X(K)+X(L))/4.
      YY(3)=(Y(I)+Y(J)+Y(K)+Y(L))/4.
C
C           FORM QUADRILATERAL STIFFNESS MATRIX  (15X15)
C
      DO 120 II=1,15
      DO 120 JJ=1,15
  120 ST(II,JJ)=0.0
C
      IX(N,5)=IX(N,1)
      LM(3)=12
      DO 150 M=1,4
      LM(1)=3*M-3
      LM(2)=LM(1)+3
       IF (M.EQ.4) LM(2)=0
      II=IX(N,M)
      JJ=IX(N,M+1)
      CALL TRISTF
C
```

```
C     ADD TRIANGULAR STIFFNESS TO QUADRILATERAL STIFFNESS
C
      DO 140 I=1,3
      DO 140 J=1,3
      DO 140 K=1,3
      DO 140 L=1,3
      II=LM(K)+I
      JJ=LM(L)+J
      LL=3*J-3+L
      KK=3*I-3+K
  140 ST(II,JJ)=ST(II,JJ)+S(KK,LL)
C
C     FORM COEFFICIENT-DISPLACEMENT TRANSFORMATION MATRIX
C
      DO 145 J=1,3
      JJ=LM(J)/3 + 1
      DO 145 I=1,3
  145 DT(I,JJ)=DT(I,JJ)+DD(I,J)
C
  150 CONTINUE
      IX(N,5)=MTYPE
C
C     ELIMINATE CENTRAL UNKNOWNS
C
      DO 160 KK=1,3
      IH=15-KK
      ID=IH+1
      DO 160 II=1,IH
      ST(II,ID)=ST(II,ID)/ST(ID,ID)
      DO 160 JJ=1,IH
  160 ST(II,JJ)=ST(II,JJ)-ST(II,ID)*ST(ID,JJ)
C
      RETURN
      END



      SUBROUTINE OUTPUT
C
      COMMON NUMNP,NUMEL,NUMMAT,NUMHAR,XHAR,ZL,PI,HED(12),C11,C12,G
     1 ,MBAND,NUMBLK,B(108),A(108,54)
      COMMON /ELEARG/ IX(300,5),E(8,3,12),NUMTC(12),T(400),SIG(300,6)
      COMMON /ORDARG/ X(350),Y(350),UX(350),UY(350),UZ(350),CODE(350)
C
      DO 50 N=1,NUMEL
      CODE(N)=0.0
      T(N)=0.0
      DO 50 I=1,4
      II=IX(N,I)
      T(N)=T(N)+X(II)/4.
   50 CODE(N)=CODE(N)+Y(II)/4.
C
      NUMST=6*NUMEL
      NUMUK=3*NUMNP
  100 REWIND 4
C
```

```
C
      READ  (5,1000) ZZ
C
      DO 550 N=1,NUMNP
      UX(N)=0.0
      UY(N)=0.0
  550 UZ(N)=0.0
C
      DO 675 N=1,NUMEL
      DO 675 I=1,6
  675 SIG(N,I)=0.0
C
      DO 800 NHAR=1,NUMHAR
      XHAR=NHAR-1
      ZX=PI*XHAR*ZZ/ZL
      SINZ=SIN(ZX)
      COSZ=COS(ZX)
C
      READ  (4) (B(N),N=1,NUMUK)
C
      DO 600 N=1,NUMNP
      UX(N)=UX(N)+B(3*N-2)*COSZ
      UY(N)=UY(N)+B(3*N-1)*COSZ
  600 UZ(N)=UZ(N)+B(3*N)*SINZ
C
      READ (4) (B(N),N=1,NUMST)
C
      DO 700 N=1,NUMEL
      SIG(N,1)=SIG(N,1)+B(6*N-5)*COSZ
      SIG(N,2)=SIG(N,2)+B(6*N-4)*COSZ
      SIG(N,3)=SIG(N,3)+B(6*N-3)*COSZ
      SIG(N,4)=SIG(N,4)+B(6*N-2)*COSZ
      SIG(N,5)=SIG(N,5)+B(6*N-1)*SINZ
  700 SIG(N,6)=SIG(N,6)+B(6*N)*SINZ
C
  800 CONTINUE
C
      WRITE (6,2001) (N,X(NN),Y(N),ZZ,UX(N),UY(N),UZ(N),N=1,NUMNP)
C .
      WRITE (6,2002) (N,T(N),CODE(N),ZZ,(SIG(N,I),I=1,6),N=1,NUMEL)
C
      GO TO 100
C
 1000 FORMAT (1F10.0)
 2001 FORMAT (10H1 N.P. NO. 9X 1HX 9X 1HY 9X 1HZ 13X 2HUX 13X 2HUY 13X
     1 2HUZ /(I10,3F10.2,3E15.7))
 2002 FORMAT (10H1  EL. NO. 9X 1HX 9X 1HY 9X 1HZ 12X 3HSXX 12X 3HSYY 12X
     1 3HSZZ 12X 3HSXY 12X 3HSXZ 12X 3HSYZ /(I10,3F10.2,6E15.7))
C
      END
```

```
C       CALCULATE STRESS AMPLITUDES
C
        DO 400 I=1,3
        SIG(N,I)=(C11-C12)*EPS(I) + C12*(EPS(1)+EPS(2)+EPS(3))
  400   SIG(N,I+3)=G*EPS(I+3)
C
  500   CONTINUE
C
        WRITE (4) ((SIG(N,I),I=1,6),N=1,NUMEL)
C
        RETURN
        END



        SUBROUTINE BANSOL
C
        COMMON /BANARG/ MM,NUMBLK,B(108),A(108,54)
C
        NN=54
        NL=NN+1
        NH=NN+NN
        REWIND 1
        REWIND 2
        NB=0
        GO TO 150
C*********************************************************************
C       REDUCE EQUATIONS BY BLOCKS
C*********************************************************************
C
C       1. SHIFT BLOCK OF EQUATIONS
C
  100   NB=NB+1
        DO 125 N=1,NN
        NM=NN+N
        B(N)=B(NM)
        B(NM)=0.0
        DO 125 M=1,MM
        A(N,M)=A(NM,M)
  125   A(NM,M)=0.0
C
C       2. READ NEXT BLOCK OF EQUATIONS INTO CORE
C
        IF (NUMBLK-NB) 150,200,150
  150   READ (2) (B(N),(A(N,M),M=1,MM),N=NL,NH)
        IF (NB) 200,100,200
```

```
C
C       3. REDUCE BLOCK OF EQUATIONS
C
  200 DO 300 N=1,NN
      IF (A(N,1)) 225,300,225
  225 B(N)=B(N)/A(N,1)
      DO 275 L=2,MM
      IF (A(N,L)) 230,275,230
  230 C=A(N,L)/A(N,1)
      I=N+L-1
      J=0
      DO 250 K=L,MM
      J=J+1
  250 A(I,J)=A(I,J)-C*A(N,K)
      B(I)=B(I)-A(N,L)*B(N)
      A(N,L)=C
  275 CONTINUE
  300 CONTINUE
C
C       4. WRITE BLOCK OF REDUCED EQUATIONS ON TAPE 1
C
      IF (NUMBLK-NB) 375,400,375
  375 WRITE (1) (B(N),(A(N,M),M=2,MM),N=1,NN)
      GO TO 100
C*********************************************************************
C     BACK-SUBSTITUTION
C*********************************************************************
  400 DO 450 M=1,NN
      N=NN+1-M
      DO 425 K=2,MM
      L=N+K-1
  425 B(N)=B(N)-A(N,K)*B(L)
      NM=N+NN
      B(NM)=B(N)
  450 A(NM,NB)=B(N)
      NB=NB-1
      IF (NB) 475,500,475
  475 BACKSPACE 1
      READ (1) (B(N),(A(N,M),M=2,MM),N=1,NN)
      BACKSPACE 1
      GO TO 400
C*********************************************************************
C     ORDER UNKNOWNS IN B ARRAY
C*********************************************************************
  500 K=0
      DO 600 NB=1,NUMBLK
      DO 600 N=1,NN
      NM=N+NN
      K=K+1
  600 B(K)=A(NM,NB)
C
      RETURN
C
      END
```